# $Id: PKGBUILD 89815 2013-05-02 15:11:19Z alucryd $
# Maintainer: Maxime Gauduin <alucryd at gmail dot com>
# Contributor : sebikul <sebikul@gmail.com>

pkgname=sqlheavy
pkgver=0.1.1
pkgrel=6
pkgdesc="GObject SQLite wrapper"
arch=('i686' 'x86_64')
url="https://code.google.com/p/sqlheavy/"
license=('LGPL')
conflicts=('sqlheavy-git')
depends=('gtk2' 'sqlite3')
makedepends=('gobject-introspection' 'vala')
options=('!libtool')
source=("https://sqlheavy.googlecode.com/files/${pkgname}-${pkgver}.tar.xz")
sha256sums=('e8689f5fd4e0baf98eada2b3811262bb99877c2e8586fd21ad2a7ad3acc59031')

prepare() {
  cd "${srcdir}"/${pkgname}-${pkgver}

  sed -i 's/libvala-0.16/libvala-0.20/' configure
  sed -i '/examples/d
          /utils/d'  Makefile.in
}

build() {
  cd "${srcdir}"/${pkgname}-${pkgver}

  ./configure --prefix=/usr --disable-static
  make
}

package() {
  cd "${srcdir}"/${pkgname}-${pkgver}

  make DESTDIR="${pkgdir}" install
}

# vim: ts=2 sw=2 et:
